Math 421 • Fall 2010

Birth of complex numbers: cubic equations

27 August 2010

Prerequisites

Mathematica

Aside from having a working *Mathematica* system at your disposal and knowing how to type input, how to evaluate an Input cell, and how to navigate around a notebook, there are really no prerequisites.

In fact, working through this notebook is a good way to learn some *Mathematica* basics and even some more advanced *Mathematica* techniques.

All the input shown uses the display form ⅈ obtained by typing ESC `ii` ESC, but everything would work identically if you typed `I` instead.

David Park's *Presentations* application is ***not*** needed for this notebook.

Mathematics

Aside from basic algebra and geometry, there are no particular mathematical prerequites. You should know the quadratic formula for solving a quadratic equation. It's helpful if you remember the Binomial Formula for expanding a power $(a+b)^n$, but the relevant case here is $n = 3$, and the formula is given explicitly in that case.

The problem

The problem is to solve the general cubic equation:

$$x^3 + a x^2 + b x + c = 0$$

```
In[1]:= cubic = x^3 + a x^2 + b x + c
Out[1]= c + b x + a x^2 + x^3
```

Mathematica can solve it directly:

In[2]:=
```
x /. Solve[cubic == 0, x]
```

Out[2]=

$$\left\{-\frac{a}{3}-\left(2^{1/3}\left(-a^2+3b\right)\right)\Big/\left(3\left(-2a^3+9ab-27c+3\sqrt{3}\sqrt{-a^2b^2+4b^3+4a^3c-18abc+27c^2}\right)^{1/3}\right)+\frac{1}{3\times 2^{1/3}}\left(-2a^3+9ab-27c+3\sqrt{3}\sqrt{-a^2b^2+4b^3+4a^3c-18abc+27c^2}\right)^{1/3},\right.$$

$$-\frac{a}{3}+\left(\left(1+i\sqrt{3}\right)\left(-a^2+3b\right)\right)\Big/\left(3\times 2^{2/3}\left(-2a^3+9ab-27c+3\sqrt{3}\sqrt{-a^2b^2+4b^3+4a^3c-18abc+27c^2}\right)^{1/3}\right)-\frac{1}{6\times 2^{1/3}}\left(1-i\sqrt{3}\right)\left(-2a^3+9ab-27c+3\sqrt{3}\sqrt{-a^2b^2+4b^3+4a^3c-18abc+27c^2}\right)^{1/3},$$

$$\left.-\frac{a}{3}+\left(\left(1-i\sqrt{3}\right)\left(-a^2+3b\right)\right)\Big/\left(3\times 2^{2/3}\left(-2a^3+9ab-27c+3\sqrt{3}\sqrt{-a^2b^2+4b^3+4a^3c-18abc+27c^2}\right)^{1/3}\right)-\frac{1}{6\times 2^{1/3}}\left(1+i\sqrt{3}\right)\left(-2a^3+9ab-27c+3\sqrt{3}\sqrt{-a^2b^2+4b^3+4a^3c-18abc+27c^2}\right)^{1/3}\right\}$$

(How did it do that?)

The output from *Mathematica* above is a list of three solutions. Notice that these solutions seem to involve non-real complex numbers—numbers of the form $\alpha + \beta\, i$ where α and β are real. As you may know, a cubic equation has three solutions—either three real solutions or else one real solution and a pair of non-real complex-conjugate solutions. So for particular coefficients a, b, c, even the solutions above that explicitly involve the complex number i actually simplify to real numbers.

Exercise 1. The cubic equation $x^3 - 5x^2 + 5x + 3 = 0$ has $x = 3$ as one of its solutions. Without using the *Mathematica* solution, above, find the other two. (*Hint*: If you have one solution r of a cubic equation, you may find the others by dividing the cubic polynomial by $x - r$ and then applying the quadratic formula to the resulting quadratic.)

Exercise 2. (a) The cubic equation $x^3 - 12\,x + 16 = 0$ has $x = -4$ as one of its solutions. Find the others.

(b) The cubic equation $x^3 - 6\,x^2 + 12\,x - 8 = 0$ has $x = 2$ as one of its solutions. Find the others.

(*Rhetorical question*: How must the assertion above, “a cubic equation has three solutions” be interpreted?

Exercise 3. Use *Mathematica* to find all solutions of the cubic equation:

(a) $x^3 - 5\,x^2 + 5\,x + 3 = 0$.

(b) $x^3 - 4\,x^2 + 14\,x - 20 = 0$.

(c) $4\,x^3 - 16\,x^2 + 4\,x + 24 = 0$.

Strategy

The strategy for finding a solution of a cubic equation considered here is:

- reduce the cubic to a “depressed cubic”—one with no quadratic term—by making a linear substitution (Cardano’s method);
- obtain one solution r of the depressed cubic from a formula of del Ferro and Tartaglia;
- when that solution r involves $\sqrt{-1}$, use Bombelli’s method to obtain a real solution;
- find the corresponding root of the original cubic by reversing the linear substitution; and
- find the other two roots of the cubic by factoring out $x - r$ and applying the quadratic formula.

Cardano’s method: reduction of cubic to depressed cubic

The following method for changing the form of a cubic was described by Girolamo Cardano in his book *Ars Magna*, 1545, but invented at the end of the 14th century by some unknown mathematician.

Cardano’s method: Make the linear substitution

$$x \to x - \frac{1}{3}\,a$$

In[3]:=
```
cubic
depressed = cubic /. x → x - (1/3) a
```

Out[3]= $c + b\,x + a\,x^2 + x^3$

Out[4]= $c + b\left(-\frac{a}{3} + x\right) + a\left(-\frac{a}{3} + x\right)^2 + \left(-\frac{a}{3} + x\right)^3$

Collect coefficients of the powers of x:

In[5]:=
```
Collect[depressed, x]
```

Out[5]= $\frac{2\,a^3}{27} - \frac{a\,b}{3} + c + \left(-\frac{a^2}{3} + b\right) x + x^3$

That cubic, which has no x^2 term, is said to be “depressed”. Write it in the form:

> ***Depressed cubic***:
>
> $$x^3 + 3\,p\,x + 2\,q$$

(With the coefficients of the depressed cubic written as multiples $3\,p$ and $2\,q$, subsequent formulas become simpler, as you will see.)

Express the values of p and q in the depressed cubic in terms of a, b, c by comparing the corresponding coefficients:

In[6]:=
```
niceDepressed = x^3 + 3 p x + 2 q;
CoefficientList[niceDepressed, x]
CoefficientList[depressed, x]
```

Out[7]= $\{2\,q,\ 3\,p,\ 0,\ 1\}$

Out[8]= $\left\{\frac{2\,a^3}{27} - \frac{a\,b}{3} + c,\ -\frac{a^2}{3} + b,\ 0,\ 1\right\}$

Thus

$$p = -\frac{a^2}{9} + \frac{b}{3}, \qquad -\frac{a^3}{27} - \frac{a\,b}{6} + \frac{c}{2}.$$

> **Exercise 4.** By hand, calculate the depressed cubic obtained by Cardano’s method for the cubic equation $x^3 + 6\,x^2 - 5\,x + 11 = 0$. Repeat for the cubic equation $x^3 - 12\,x^2 + 2\,x - 4 = 0$.

Exercise 5. Suppose you had used Cardano's method to obtain the depressed cubic $x^3 + 8\,x + 5$ from a cubic $x^3 + a\,x^2 + b\,x + c$. If $a = -9$, what was the original cubic?

Exercise 6. Suppose you had used Cardano's method to obtain a depressed cubic $x^3 + 3\,p\,x + 2\,q$ from a cubic $x^3 + a\,x^2 + b\,x + c$, where $a = -9$. One of the roots of the depressed cubic is $x = 5$. What is the corresponding root of the original cubic?

Exercise 7. Suppose you wrote the depressed cubic in the form $x^3 + \beta\,x + \gamma$, without the coefficient multipliers of 3 and 2. Express the coefficients β and γ in terms of the original cubic's coefficients a, b, c.

Exercise 8. The linear substitution used was $x \to x - \frac{1}{3}\,a$. Among all possible linear substitutions $x \to x - \text{cst}$, why use cst $= \frac{1}{3}\,a$?

The del Ferro–Tartaglia formula

In 1515 Scipione del Ferro discovered a formula, which he kept secret, for finding a root of a depressed cubic $x^3 + 3\,p\,x + 2\,q$ in terms of p and q. In 1530 Niccolò Fontana, aka "Tartaglia", revealed the same formula to Cardano.

In *Mathematica*, the formula is given by the function definition:

In[9]:=
```
delFerroTartagliaRoot[p_, q_] :=
```
$$\left(-q + \sqrt{p^3 + q^2}\right)^{1/3} + \left(-q - \sqrt{p^3 + q^2}\right)^{1/3}$$

For example:

In[10]:=
```
delFerroTartagliaRoot[-5, -2]
```

Out[10]=
```
4
```

Exercise 9. Use the del Ferro-Tartaglia formula by hand to calculate a root of the depressed cubic $x^3 - 9\,x + 8$. (You may leave your answer in a form involving cube-roots.) Check your answer against the result of using the *Mathematica* function `delFerroTartagliaRoot`.

Exercise 10. Apply Cardano's method and then the del Ferro-Tartaglia formula to find a root of the cubic $x^3 + 15\,x^2 + 57\,x + 27$. (You may leave your answer in a form involving cube-roots.)

Exercise 11. Suppose you wrote the depressed cubic in the form $x^3 + \beta\,x + \gamma$, without the coefficient multipliers of 3 and 2. What now would the del Ferro-Tartaglia formula for a root be, in terms of β and γ?

(Rhetorical question: Do you see now why the del Ferro-Tartaglia formula is simpler when the multipliers are included in the coefficients of the depressed cubic?)

The del Ferro Tartaglia formula for solving a depressed cubic equation $x^2 + 3\,p\,x + 2\,q = 0$ can readily be converted into formulas for solving cubic equations of the form $x^3 + 3\,p\,x = 2\,q$, $x^3 + 2\,q = 3\,p\,x$, and $x^3 = 3\,p\,x + 2\,q$. We mention this because, in the work of del Ferro and Tartaglia, p and q had to be positive numbers: like most European mathematicians of their time, they did not accept the notion of negative numbers.

How did del Ferro and Tartaglia devise their formula? We don't know, but here's a possible way, which uses reasoning by analogy.

As they knew, as as you can readily check,

$$x = \sqrt{a + \sqrt{b}} + \sqrt{a - \sqrt{b}}$$

is a solution of the quadratic equation

$$x^2 = 2\sqrt{a^2 - b} + 2\,a.$$

when $a > \sqrt{b}$ and $b > 0$.

Exercise 12. Verify the statement made above about the solution of $x^2 = 2\sqrt{a^2 - b} + 2\,a$.

Then perhaps, by analogy,

$$x = \sqrt[3]{a + \sqrt{b}} + \sqrt[3]{a - \sqrt{b}}$$

is a solution of the cubic equation

$$x^3 = 3\left(\sqrt[3]{a^2 - b}\right)x + 2\,a.$$

If in the latter, cubic, equation you take $p = -\sqrt[3]{a^2 - b}$ and $q = -a$, then $p^3 + q^2 = b$. Thus the guess for the solution of this cubic equation is indeed what the del Ferro-Tartaglia formula gives.

Exercise 13. Verify the statement made above about the solution of $x^3 = 3\left(\sqrt[3]{a^2 - b}\right)x + 2\,a$.

A paradox

Fact 1: a depressed cubic *always* has a *real* solution.

In fact,

$$x^3 + 3\,p\,x + 2\,q = 0$$

is equivalent to

$$x^3 = -3\,p\,x - 2\,q,$$

and the graph of the cube function

$$y = x^3$$

always intersects the line

$$y = -3\,p\,x - 2\,q$$

in at least one point, no matter what p and q are!

The following plots provide evidence to support Fact 1.

In[11]:=
```
Plot[{x^3, 15 x + 4}, {x, -5, 5}]
```

Out[11]=



In[12]:=
```
Manipulate[
 Plot[{x^3, -3 p x - 2 q}, {x, -8, 8}, PlotRange → {-500, 500},
  PlotStyle → {Red, Blue}, AxesLabel → {x, y}],
 {p, -20, 20, Appearance → "Labeled"}, {q, -20, 20, Appearance → "Labeled"}]
```

Out[12]=



Move the sliders for p and q in the output above to see where the two curves meet.

> **Exercise 14.** Explain in detail why, in fact, the curve $y = x^3$ must actually intersect the line $y = -3\,p\,x - 2\,q$ in at least one point (x, y), no matter what the values of p and q are.

Fact 2: del Ferro-Tartaglia formula involves *square-roots of negative numbers* if $q^2 < -p^3$.

In fact, recall the formula is the value:

In[13]:=
```
delFerroTartagliaRoot[p, q]
```

Out[13]=

$$\left(-q - \sqrt{p^3 + q^2}\right)^{1/3} + \left(-q + \sqrt{p^3 + q^2}\right)^{1/3}$$

As Cardano noticed, the square-root here may be that of a negative number. This is the case, for example, when the depressed cubic $x^3 + 3\,p\,x + 2\,q$ has $p = -5,\ q = -2$:

```
In[14]:= p^3 + q^2 /. { p → -5, q → -2}
Out[14]= -121
```

Resolving the paradox

Bombelli’s method, explained next, resolves the paradox: it changes the form of the root provided by the del Ferro-Tartaglia formula so as to see it is actually real.

Bombelli’s method

In his book *L’algebra*, 1572, Rafael Bombelli looked at the depressed cubic $x^3 + 3\,p\,x + 2\,q$ when $p = -5$ and $q = -2$:

$$x^3 - 15\,x - 4 = 0$$

```
In[15]:= example = niceDepressed /. {p → -5, q → -2}
Out[15]= -4 - 15 x + x^3
```

Of course, there *must* be at least one solution—the cubic $y = x^3$ and the line $y = 15\,x + 4$ must intersect:

```
In[16]:= Plot[{x^3, 15 x + 4}, {x, -7, 7}, PlotStyle → {Red, Blue}]
```

Out[16]=



In fact, the plot suggests that $x = 4$ is a solution. Verify that it really is:

In[17]:=
```
example /. x → 4
```
Out[17]= 0

That graphical approach is not what Bombelli used. Instead, he used an algebraic approach.

Bombelli's "wild thought"

The del Ferro-Tartaglia formula for a root is:

In[18]:=
```
delFerroTartagliaRoot[p, q]
```
Out[18]= $\left(-q-\sqrt{p^3+q^2}\right)^{1/3}+\left(-q+\sqrt{p^3+q^2}\right)^{1/3}$

And in the example, the quantity under the square-root signs is:

In[19]:= p^3+q^2 /. {p → -5, q → -2}

Out[19]= -121

So the del Ferro-Tartaglia solution in this example is:

$$\sqrt[3]{2+\sqrt{-121}}+\sqrt[3]{2-\sqrt{-121}}=\sqrt[3]{2+11\sqrt{-1}}+\sqrt[3]{2-11\sqrt{-1}}$$

Bombelli’s “wild thought” was the following. *Assume* there are numbers *m* and *n* with:

$$\sqrt[3]{2+11\sqrt{-1}}=m+n\sqrt{-1},\quad \sqrt[3]{2-11\sqrt{-1}}=m-n\sqrt{-1}. \qquad (*)$$

Then the sum $\sqrt[3]{2+11\sqrt{-1}}+\sqrt[3]{2-11\sqrt{-1}}$ would equal:

In[20]:= $\left(m+n\sqrt{-1}\right)+\left(m-n\sqrt{-1}\right)$

Out[20]= 2 m

Thus to obtain a (real) solution of the depressed cubic, Bombelli just needs to solve (*) for *m* and *n*. And that amounts to solving:

$$\left(m+n\sqrt{-1}\right)^3=2+11\sqrt{-1},\quad \left(m-n\sqrt{-1}\right)^3=2-11\sqrt{-1} \qquad (**)$$

How? Use the binomial formula to expand $\left(m+n\sqrt{-1}\right)^3$, *assuming*:

- the usual rules of algebra hold for expressions involving

$$i=\sqrt{-1}$$

- the special rule

$$i^2 = \left(\sqrt{-1}\right)^2 = -1$$

Exercise 15. Use the binomial formula for cubes, $(x + y)^3 = x^3 + 3\,x^2\,y + 3\,x\,y^2 + y^3$, along with the rules about $\sqrt{-1}$ assumed above, in order to express $\left(m + n\sqrt{-1}\right)^3$ in the form $u + v\sqrt{-1}$ for real u and v. Of course u and v each be an expression in terms of m and n.

Mathematica already knows the binomial formula as well as the rules assumed for algebraic expressions involving the "imaginary" number $i = \sqrt{-1}$:

```
In[21]:= theCube = Expand[(m + n √-1)^3]

Out[21]= m^3 + 3 i m^2 n - 3 m n^2 - i n^3
```

Separate the "real part" from the "imaginary part" that multiplies i:

```
In[22]:= ComplexExpand[theCube]

Out[22]= m^3 - 3 m n^2 + i (3 m^2 n - n^3)
```

See the Appendix for a discussion of `ComplexExpand`.

Thus the desired m and n are to satisfy the complex equation:

$$m^3 - 3\,m\,n^2 + i\left(3\,m^2\,n - n^3\right) = 2 + 11\,i$$

That left-hand side's real and imaginary parts are…

```
In[23]:= cubeParts = ComplexExpand[{Re[theCube], Im[theCube]}]

Out[23]= {m^3 - 3 m n^2, 3 m^2 n - n^3}
```

…and according to the first equation of (**), those parts should equal

```
In[24]:= {Re[2 + 11 i], Im[2 + 11 i]}

Out[24]= {2, 11}
```

Thus solving the first equation of (**) amounts to solving:

```
In[25]:= equations =
          cubeParts == {2, 11}

Out[25]= {m^3 - 3 m n^2, 3 m^2 n - n^3} == {2, 11}
```

Write that as two separate scalar equations:

In[26]:=
```
equations = Thread[equations]
```
Out[26]=
```
{m^3 - 3 m n^2 == 2, 3 m^2 n - n^3 == 11}
```

Factor the left-hand sides:

In[27]:=
```
equations = Factor[equations]
```
Out[27]=
```
{m (m^2 - 3 n^2) == 2, n (3 m^2 - n^2) == 11}
```

Now Bombelli seeks solutions m and n of the equations that are positive *integers*. The process is indicated in the following exercise.

Exercise 16. By hand, and without merely guessing, find the integers m and n that are solutions of the pair of equations

$$\begin{cases} m\left(m^2-3\,n^2\right) &= 2, \\ n\left(3\,m^2-n^2\right) &= 11. \end{cases}$$

(*Hint.* The only positive integer factors of 2 are 1 and 2. From the first equation, this means that either $m = 1$ or else $m = 2$. Show why the case $m = 1$ is impossible, so that $m = 2$. Then use the second equation to determine n. Be sure to verify that your m and n actually satisfy both equations.)

Mathematica can find integer solutions of the equations directly:

In[28]:=
```
Reduce[equations, {m, n}, Integers]
```
Out[28]=
```
m == 2 && n == 1
```

Exercise 17. Check that the solution found for the first equation in (**) also satisfies the second equation.

Thus Bombelli's solution of the depressed cubic $x^3 - 15\,x - 4 = 0$ of his example is $x = 4$.

Exercise 18. Go through Bombelli's process of solution for his example, but start with the second equation in (**) instead of the first.

Exercise 19. Use *Mathematica* to find *all* solutions of `equations` in terms of `m` and `n`. Use those to find the corresponding values of $2\,m$. Which of those values are (not necessarily real) solutions of the depressed cubic equation $x^3 - 15\,x - 4 = 0$ and which are not?

Exercise 20. Find a real root of the given depressed cubic equation by applying Bombelli's method to the result from the del Ferro-Tartaglia formula:

(a) $x^3 - 6x + 4 = 0$.

(b) $x^3 - 102x + 20 = 0$.

Exercise 21. Find a real root of the given cubic equation by first applying Cardano's method to obtain a depressed cubic and then proceeding as in the preceding exercise. Be sure your final answer is a root of the given cubic rather than a root of the depressed cubic!

(a) $x^3 + 6x^2 - 18x - 88 = 0$.

(b) $x^3 + 2x^2 - (176/3)x - 1936/27 = 0$.

Exercise 22. Find *all* solutions of each of the cubic equations in the preceding exercise. (*Hint*: If you have one root r of a cubic, you may find the others by dividing the cubic by $x - r$ and then applying the quadratic formula to the resulting quadratic.)

Exercise 23. The "usual rules of algebra" include such identities as:

$(x + y) + (u + v) = (x + u) + (y + v)$,

$(x + y)(u + v) = xu + yv + xv + yu$,

$k(x + y) = kx + ky$,

$k(xy) = (kx)y = x(ky)$,

$(x + y) + z = x + (y + z)$.

You know that these identities do hold for real numbers x, y, u, v, k, z.

Assume that such identities also hold for "complex numbers"—numbers of the form $a + bi$ where a and b are real. And continue to assume that $i^2 = i\,i = -1$. Then put each of the following into the form $u + iv$ with u and v real:

$(a + bi) + (c + di)$, $(a + bi)(c + di)$

The moral

Square-roots of negative numbers are useful in obtaining *real* roots of certain cubic equations.

(But what *are* such "complex" numbers? That's what's next in this course!)

Appendix: real and imaginary parts

To find the real and imaginary parts of $2 + 11\,i$ with *Mathematica*, directly use **Re** and **Im**:

In[29]:=
```
{Re[2 + 11 ⅈ], Im[2 + 11 ⅈ]}
```

Out[29]=
```
{2, 11}
```

But trying the same thing directly with…

In[30]:=
```
theCube = Expand[(m + n √-1)^3]
```

Out[30]= $m^3 + 3\,i\,m^2\,n - 3\,m\,n^2 - i\,n^3$

…will *not* work:

In[31]:=
```
{Re[theCube], Im[theCube]}
```

Out[31]= $\{-3\,\mathrm{Im}[m^2\,n] + \mathrm{Im}[n^3] + \mathrm{Re}[m^3 - 3\,m\,n^2],\ \mathrm{Im}[m^3 - 3\,m\,n^2] + 3\,\mathrm{Re}[m^2\,n] - \mathrm{Re}[n^3]\}$

The reason is that, **by default, *Mathematica* regards all symbolic variables representing numbers to be complex**!

You have to explicitly tell *Mathematica* when you want all such variables in an expression, instead, to be regarded as real. And to do that, you use **ComplexExpand**.

For example, as was done earlier:

In[32]:=
```
ComplexExpand[theCube]
```

Out[32]= $m^3 - 3\,m\,n^2 + i\,(3\,m^2\,n - n^3)$

In[33]:=
```
ComplexExpand[{Re[theCube], Im[theCube]}]
```

Out[33]= $\{m^3 - 3\,m\,n^2,\ 3\,m^2\,n - n^3\}$

> **Exercise 24.** Explain why `Re[ComplexExpand[theCube]]` does not give an explicit value (in terms of `m` and `n`) for the real part of `theCube`, whereas `ComplexExpand[Re[theCube]]` does.

Sometimes you want to bring in `ComplexExpand` as an “afterthought” to the main expression, and then you may use the following “postfix” form of input:

```
In[34]:= theCube // ComplexExpand
         {Re[theCube], Im[theCube]} // ComplexExpand
```

Out[34]= $m^3 - 3\,m\,n^2 + i\,(3\,m^2\,n - n^3)$

Out[35]= $\{m^3 - 3\,m\,n^2,\ 3\,m^2\,n - n^3\}$

Then the desired values of *m* and *n* in Bombelli's example are given by:

```
In[36]:= First@Solve[ComplexExpand[{Re[theCube], Im[theCube]}] ==
            {Re[2 + 11 i], Im[2 + 11 i]}, {m, n}]
```

Out[36]= {m → 2, n → 1}

(The reason for using `First` like that is to discard all but the first, real, solution.)

Mathematica will *not* provide actual values for *m* and *n* that are solutions of $(m + n\,i)^3 = 2 + 11\,i$ if you try the following:

```
In[37]:= Solve[ComplexExpand[theCube] == 2 + 11 i, {m, n}]
```

Solve::svars : Equations may not give solutions for all "solve" variables. ≫

Out[37]= $\left\{\left\{m \to \frac{1}{2}\,i\left((-1 + 2\,i) + (2 + i)\,\sqrt{3} - 2\,n\right)\right\},\ \{m \to (2 + i) - i\,n\},\ \left\{m \to -\frac{1}{2}\,i\left((1 - 2\,i) + (2 + i)\,\sqrt{3} + 2\,n\right)\right\}\right\}$

Rather, as you see, it just expresses one of the variables in $(m + n\,i)^3 = 2 + 11\,i$ in terms of the other.

If you want *Mathematica* to determine actual values for *m* and *n* (without specifying that they be integers), you need to bring in the second equation, $(m - n\,i)^3 = 2 - 11\,i$, too.

And then there is no need to separate each complex equation into a pair of real equations, one for real parts and the other for imaginary parts; *Mathematica* can handle the entire solution in one fell swoop:

```
In[38]:= First@Solve[{(m + n i)^3 == 2 + 11 i, (m - n i)^3 == 2 - 11 i}, {m, n}]
```

Out[38]= {m → 2, n → 1}

Exercise 25. The *Mathematica* function `Conjugate` changes a complex number $a + b\,i$ into $a - b\,i$ when *a* and *b* are real. For example, input `Conjugate[2 + 11 i]` gives output `2 - 11 i`.

Why doesn't the input

```
Conjugate[m + n i]
```

give output `m - n i`? And how can you modify that input so as to obtain output `m - n i`?

For more information about `ComplexExpand`, see notebook `CartesianPolarForms.nb`.

References

For a summary of the history of solving cubics, see the article “Cubic function”, *Wikipedia*, http://en.wikipedia.org/wiki/Cubic_function.